Code used in BPPM model spreadsheet

This Mathematica notebook generates the code is used in the BPPM model spreadsheet.

BPPM model equations

The BPPM modeling equations are symbolically solved for the variables of interest. Here are the design equations to be solved.

```
substitutions = {
    dehc -> hc lambda D0^2,
    M -> 2^m,
    BPP -> (1 - Exp[-KsR]) m / KsR
   };
BPPMmodel = {
    LambdaPeakS == PpT AeT AeS / dehc,
    KxS == effic Fx LambdaPeakS TF,
    KsS == effic LambdaPeakS Ts,
    KnS == effic We LambdaNS TF,
    KiS == effic We Fc AeS LambdaIS TF,
    KdS == LambdaDS TF,
    KsS == SBR (KxS + KnS + KiS + KdS),
    SXR == KsS / KxS,
    SNR == KsS / KnS,
    SIR == KsS / KiS,
    SDR == KsS / KdS,
    (* maximum feasible value of SBR corresponding to NS=1 *)
    SBRmax == KsR / (KxS + KnS + KiS + KdS),
    PpT Ts == PaT TI,
    TF == M Ts,
    (* OmegaA assumed to have units of arcsec^2, asToSr converts to steradians *)
    AeS (asToSr OmegaA) == lambda^2,
    R TI == KsR BPP,
    δ == TF / TI,
    1 - PO == (1 - PD) (1 - PW),
    Ractual == (1 - PO) R,
    NS KsS == KsR
   };
outputs = {
    PaT, PpT, KsS, KxS, KnS, KiS, KdS, SXR, SNR, SIR, SDR,
    SBRmax,
    LambdaPeakS, TF, AeS, TI,
    δ, PO, Ractual, NS
   };
(* substitute for BPP and M after Solve[],
because otherwise Solve[] has difficulty *)
solutions = Solve[BPPMmodel, outputs, Reals][[1]] /. substitutions // Simplify;
```

The following are the convenient formatted solutions for parameters and metrics of interest.

```
solutions // TableForm
```

Out[29]//TableForm=

$$\text{PaT} \to -\frac{2^m \,\text{D0}^2\, e^{\text{KsR}}\, \text{hc R SBR Ts}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}{\text{AeT}\, (-1 + e^{\text{KsR}})\, \text{effic lambda m}\, (-1 + 2^m \text{ Fx SBR})}$$

$$\text{PpT} \to -\frac{2^m \,\text{D0}^2\, \text{hc SBR}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}{\text{AeT effic lambda}\, (-1 + 2^m \text{ Fx SBR})}$$

$$\text{KsS} \to -\frac{2^m \,\text{SBR Ts}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}{\text{asToSr OmegaA}\, (-1 + 2^m \text{ Fx SBR})}$$

$$\text{KxS} \to -\frac{4^m \,\text{Fx SBR Ts}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}{\text{asToSr OmegaA}\, (-1 + 2^m \text{ Fx SBR})}$$

$$\text{KnS} \to 2^m \text{ effic LambdaNS Ts We}$$

$$\text{KiS} \to \frac{2^m \text{ effic Fc lambda}^2 \text{ LambdaIS Ts We}}{\text{asToSr OmegaA}}$$

$$\text{KdS} \to 2^m \text{ LambdaDS Ts}$$

$$\text{SXR} \to \frac{2^{-m}}{\text{Fx}}$$

$$\text{SNR} \to \frac{\text{asToSr LambdaDS OmegaA SBR} + \text{effic Fc lambda}^2 \text{ LambdaIS SBR We} + \text{asToSr effic LambdaNS OmegaA SBR We}}{\text{asToSr effic LambdaNS OmegaA We} - 2^m \text{ asToSr effic Fx LambdaNS OmegaA SBR We}}$$

$$\text{SIR} \to \frac{\text{asToSr LambdaDS OmegaA SBR} + \text{effic Fc lambda}^2 \text{ LambdaIS SBR We} + \text{asToSr effic LambdaNS OmegaA SBR We}}{\text{effic Fc lambda}^2 \text{ LambdaIS We} - 2^m \text{ effic Fc Fx lambda}^2 \text{ LambdaIS SBR We}}$$

$$\text{SDR} \to \frac{\text{asToSr LambdaDS OmegaA SBR} + \text{effic Fc lambda}^2 \text{ LambdaIS SBR We} + \text{asToSr effic LambdaNS OmegaA SBR We}}{\text{asToSr LambdaDS OmegaA} - 2^m \text{ asToSr Fx LambdaDS OmegaA SBR}}$$

$$\text{SBRmax} \to \frac{\text{asToSr KsR OmegaA}\, (2^{-m} - \text{Fx SBR})}{\text{Ts}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}$$

$$\text{LambdaPeakS} \to -\frac{2^m \text{ SBR}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}{\text{asToSr effic OmegaA}\, (-1 + 2^m \text{ Fx SBR})}$$

$$\text{TF} \to 2^m \text{ Ts}$$

$$\text{AeS} \to \frac{\text{lambda}^2}{\text{asToSr OmegaA}}$$

$$\text{TI} \to \frac{\text{m} - e^{-\text{KsR}}\, \text{m}}{\text{R}}$$

$$\delta \to \frac{2^m\, e^{\text{KsR}} \text{ R Ts}}{(-1 + e^{\text{KsR}})\, \text{m}}$$

$$\text{PO} \to \text{PD} + \text{PW} - \text{PD PW}$$

$$\text{Ractual} \to (-1 + \text{PD})\, (-1 + \text{PW})\, \text{R}$$

$$\text{NS} \to \frac{\text{asToSr KsR OmegaA}\, (2^{-m} - \text{Fx SBR})}{\text{SBR Ts}\, (\text{effic Fc lambda}^2 \text{ LambdaIS We} + \text{asToSr OmegaA}\, (\text{LambdaDS} + \text{effic LambdaNS We}))}$$

The following unformatted versions of the design equations are useful for pasting into spreadsheets or programs.

solutions // InputForm // TableForm

Out[30]//TableForm=

```
{PaT ->
  -((2^m*D0^2*E^KsR*hc*R*SBR*Ts*(effic*Fc*lambda^2*LambdaIS*We + asToSr*
         OmegaA*(LambdaDS + effic*LambdaNS*We)))/
     (AeT*(-1 + E^KsR)*effic*lambda*m*(-1 + 2^m*Fx*SBR))),
 PpT -> -((2^m*D0^2*hc*SBR*(effic*Fc*lambda^2*LambdaIS*We +
        asToSr*OmegaA*(LambdaDS + effic*LambdaNS*We)))/
     (AeT*effic*lambda*(-1 + 2^m*Fx*SBR))),
 KsS -> -((2^m*SBR*Ts*(effic*Fc*lambda^2*LambdaIS*We +
        asToSr*OmegaA*(LambdaDS + effic*LambdaNS*We)))/
     (asToSr*OmegaA*(-1 + 2^m*Fx*SBR))), KxS ->
  -((4^m*Fx*SBR*Ts*(effic*Fc*lambda^2*LambdaIS*We +
        asToSr*OmegaA*(LambdaDS + effic*LambdaNS*We)))/
     (asToSr*OmegaA*(-1 + 2^m*Fx*SBR))),
 KnS -> 2^m*effic*LambdaNS*Ts*We,
 KiS ->
  (2^m*effic*Fc*lambda^2*LambdaIS*Ts*We)/
   (asToSr*OmegaA),
 KdS -> 2^m*LambdaDS*Ts,
 SXR ->
  1/(2^m*Fx),
 SNR -> (asToSr*LambdaDS*OmegaA*SBR + effic*Fc*lambda^2*LambdaIS*SBR*We +
     asToSr*effic*LambdaNS*OmegaA*SBR*We)/
   (asToSr*effic*LambdaNS*OmegaA*We -
     2^m*asToSr*effic*Fx*LambdaNS*OmegaA*SBR*We),
 SIR -> (asToSr*LambdaDS*OmegaA*SBR + effic*Fc*lambda^2*LambdaIS*SBR*We +
     asToSr*effic*LambdaNS*OmegaA*SBR*We)/
   (effic*Fc*lambda^2*LambdaIS*We -
     2^m*effic*Fc*Fx*lambda^2*LambdaIS*SBR*We),
 SDR -> (asToSr*LambdaDS*OmegaA*SBR + effic*Fc*lambda^2*LambdaIS*SBR*We +
     asToSr*effic*LambdaNS*OmegaA*SBR*We)/
   (asToSr*LambdaDS*OmegaA - 2^m*asToSr*Fx*LambdaDS*OmegaA*SBR),
 SBRmax -> (asToSr*KsR*OmegaA*(2^(-m) - Fx*SBR))/
   (Ts*(effic*Fc*lambda^2*LambdaIS*We +
       asToSr*OmegaA*(LambdaDS + effic*LambdaNS*We))),
 LambdaPeakS -> -((2^m*SBR*(effic*Fc*lambda^2*LambdaIS*We +
        asToSr*OmegaA*(LambdaDS + effic*LambdaNS*We)))/
     (asToSr*effic*OmegaA*(-1 + 2^m*Fx*SBR))),
 TF -> 2^m*Ts, AeS -> lambda^2/(asToSr*OmegaA),
```

```
TI ->
  (m - m / E^KsR) / R,
δ -> (2^m * E^KsR * R * Ts) / ((-1 + E^KsR) * m),
PO -> PD + PW - PD * PW,
Ractual ->
  (-1 + PD) * (-1 + PW) * R,
NS -> (asToSr * KsR * OmegaA * (2^(-m) - Fx * SBR)) /
   (SBR * Ts * (effic * Fc * lambda^2 * LambdaIS * We +
       asToSr * OmegaA * (LambdaDS + effic * LambdaNS * We)))}
```

Numerical examples

These numerical examples are useful for checking the spreadsheet results. Here are the nominal choices for parameters.

```
(* these can always be overridden by prior substitution *)
paramNominal = {
   (* parameters governing reliabilty of data recovery *)
   m → 12,
   KsR → 0.2,
   (* physical parameters *)
   D0 -> Quantity[4.24, "ly"],
   hc -> Quantity["planck's constant"] × Quantity["c"],
   SBR → 4.,
   AeT -> Quantity[100., "cm^2"],
   effic → 1.,
   (* design parrameters *)
   R -> Quantity[1., "per sec"],
   Fx → 1. × 10^-7,
   Fc → 1. × 10^-2,
   OmegaA → 10., (* implicit units of arcsec^2 *)
   Ts → Quantity[0.1, "microsec"],
   We → Quantity[10., "MHz"], (* minimum value is 1/Ts *)
   rhoM → 1.,
   (* this value is chosen to force SDR=SNR approximately *)
   LambdaDS → 32. / Quantity["year"],
   PW → 0.1,
   PD → 0.52,
   (* conversation of arcsec^2 to steradians *)
   asToSr -> (Quantity["arcsec"] / Quantity["radians"]) ^2 // N
  };
```

```
(* representation of the background numerical parameters *)
backgroundNominal = <|
   "lambda" → <|"400" → Quantity[400., "nm"], "1000" → Quantity[1., "micron"]|>,
   "interference" -> <|"400" → 8. × 10^-10 / Quantity["m^2"],
     "1000" -> 2. × 10^-7 / Quantity["m^2"]|>,
   "zodiacal" → <|"400" → 2.0 × 10^-16, "1000" → 1.0 × 10^-14|>,
   "starField" -> <|"400" → 2.0 × 10^-16, "1000" → 1.0 × 10^-14|>,
   "daylight" → <|"400" -> 4.1 × 10^-8, "1000" -> 5. × 10^-7|>,
   "moonlight" -> <|"400" → rhoM * 1. × 10^-13, "1000" → rhoM * 1.3 × 10^-12|>
   |>;
(* test *)
backgroundNominal[[All, "400"]]
backgroundNominal[[All, "1000"]]
```

Out[11]= $\langle|$ lambda → 400. nm , interference → $8. \times 10^{-10}$ per meter^2 , zodiacal → $2. \times 10^{-16}$, starField → $2. \times 10^{-16}$, daylight → 4.1×10^{-8}, moonlight → $1. \times 10^{-13}$ rhoM $|\rangle$

Out[12]= $\langle|$ lambda → 1. microns , interference → $2. \times 10^{-7}$ per meter^2 , zodiacal → $1. \times 10^{-14}$, starField → $1. \times 10^{-14}$, daylight → $5. \times 10^{-7}$, moonlight → 1.3×10^{-12} rhoM $|\rangle$

Now we define a few useful functions.

```
(*
implement wavelength-dependent substitions
   model = association of parameters
   wavelength = "400" or "1000" (units of nanometers)
    dayNight = "daylight" or "moonlight"
     rhoM = fraction of moonlight irradiance
*)
wlVals[model_, wavelength_, dayNight_] := Module[
  {column, rho},
  column = model[[All, wavelength]];
  rho = If[dayNight == "daylight", 1., rhoM];
  {
   lambda → column["lambda"],
   LambdaIS → column["interference"],
   LambdaNS → column["zodiacal"] + column["starField"] + column[dayNight]
  }
]
(* test *)
(* note you have to input backgroundNominal before running this *)
wlVals[backgroundNominal, "400", "moonlight"]
wlVals[backgroundNominal, "1000", "moonlight"]
```

Out[14]= {lambda → 400. nm, LambdaIS → $8.\times10^{-10}$ per meter^2, LambdaNS → $4.\times10^{-16} + 1.\times10^{-13}$ rhoM}

Out[15]= {lambda → 1. microns, LambdaIS → $2.\times10^{-7}$ per meter^2,
LambdaNS → $2.\times10^{-14} + 1.3\times10^{-12}$ rhoM}

In[16]:=
```
(* for a particular model,
print the model equations for a set of specified values
 as well as the values for a set of changes and parameters
 (changes override parameters) *)
(* vars is a list of {variable name, desired units} *)
prEqns[model_, vars_] :=
 Table[Print[var[[1]], " = ", var[[1]] /. model], {var, vars}]
prVals[model_, vars_, changes_, params_] := Module[
  {val, units},
  Table[
    val = var[[1]] /. model /. changes /. params;
    (* units of "" not accepted by UnitConvert,
    so substitute the generic "Metric" *)
    units = Which[var[[2]] == "", "Metric", True, var[[2]]];
    val = UnitConvert[val, units];
    Print[var[[1]], " = ", val],
    {var, vars}
   ];
 ]
```

Multi-probe coverage in moonlight @ 400 nm

```
paramTot = Join[paramNominal,
   wlVals[backgroundNominal, "400", "moonlight"] /. paramNominal];
ofInterest = {{SBRmax, ""}, {m, ""}, {TF, "ms"}, {AeS, "cm^2"}, {TI, "s"},
   {δ, ""}, {Ractual, "s^-1"}, {PaT, "mW"}, {PaT AeT, "meters^2*Watts"},
   {PpT, "kW"}, {PpT / PaT, ""}, {SXR, ""}, {SIR, ""}, {SNR, ""}, {SDR, ""},
   {NS, ""}, {AeS NS, "km^2"}, {PaT AeT AeS NS, "meter^4*Watts"}};
prVals[solutions, ofInterest, {OmegaA → 10.}, paramTot];
```

SBRmax = 2.35138×10^{8}

m = 12

TF = 0.4096 ms

AeS = 6.80723 cm^2

TI = 2.17523 s

δ = 0.000188302

Ractual = 0.432 per second

PaT = 29.3771 mW

AeT PaT = 0.000293771 m^2W

PpT = 639.02 kW

$\frac{\text{PpT}}{\text{PaT}}$ = 2.17523×10^{7}

SXR = 2441.41

SIR = 152.527

SNR = 8.2732

SDR = 8.18585

NS = 5.87845×10^{7}

AeS NS = 0.0400159 km^2

AeS AeT NS PaT = 11.7555 m^4W

Multi-probe coverage in moonlight @ 1000 nm

```
paramTot = Join[paramNominal,
   wlVals[backgroundNominal, "1000", "moonlight"] /. paramNominal];
ofInterest = {{SBRmax, ""}, {m, ""}, {TF, "ms"}, {AeS, "cm^2"}, {TI, "s"},
   {δ, ""}, {Ractual, "s^-1"}, {PaT, "mW"}, {PaT AeT, "meters^2*Watts"},
   {PpT, "kW"}, {PpT / PaT, ""}, {SXR, ""}, {SIR, ""}, {SNR, ""}, {SDR, ""},
   {NS, ""}, {AeS NS, "km^2"}, {PaT AeT AeS NS, "meter^4*Watts"}};
prVals[solutions, ofInterest, {OmegaA → 10.}, paramTot];
```

SBRmax = 4.90893×10^6

m = 12

TF = 0.4096 ms

AeS = 42.5452 cm^2

TI = 2.17523 s

δ = 0.000188302

Ractual = 0.432 per second

PaT = 562.867 mW

AeT PaT = 0.00562867 m^2W

PpT = 12 243.7 kW

$\frac{PpT}{PaT}$ = 2.17523×10^7

SXR = 2441.41

SIR = 4.67588

SNR = 30.1418

SDR = 392.103

NS = 1.22723×10^6

AeS NS = 0.00522128 km^2

AeS AeT NS PaT = 29.3888 m^4W

Single-probe coverage in moonlight @ 400 nm

```
paramTot = Join[paramNominal,
   wlVals[backgroundNominal, "400", "moonlight"] /. paramNominal];
ofInterest = {{SBRmax, ""}, {m, ""}, {TF, "ms"}, {AeS, "cm^2"}, {TI, "s"},
   {δ, ""}, {Ractual, "s^-1"}, {PaT, "mW"}, {PaT AeT, "meters^2*Watts"},
   {PpT, "kW"}, {PpT / PaT, ""}, {SXR, ""}, {SIR, ""}, {SNR, ""}, {SDR, ""},
   {NS, ""}, {AeS NS, "km^2"}, {PaT AeT AeS NS, "meter^4*Watts"}};
prVals[solutions, ofInterest, {OmegaA → 0.01}, paramTot];
```

SBRmax = 8.63157×10^{6}

m = 12

TF = 0.4096 ms

AeS = 6807.23 cm^2

TI = 2.17523 s

δ = 0.000188302

Ractual = 0.432 per second

PaT = 0.800281 mW

AeT PaT = 8.00281×10^{-6} m^2W

PpT = 17.4079 kW

$\frac{\text{PpT}}{\text{PaT}}$ = 2.17523×10^{7}

SXR = 2441.41

SIR = 4.15508

SNR = 225.375

SDR = 222.996

NS = 2.15789×10^{6}

AeS NS = 1.46893 km^2

AeS AeT NS PaT = 11.7555 m^4W

Single-probe coverage in moonlight @ 1000 nm

```
paramTot = Join[paramNominal,
   wlVals[backgroundNominal, "1000", "moonlight"] /. paramNominal];
ofInterest = {{SBRmax, ""}, {m, ""}, {TF, "ms"}, {AeS, "cm^2"}, {TI, "s"},
   {δ, ""}, {Ractual, "s^-1"}, {PaT, "mW"}, {PaT AeT, "meters^2*Watts"},
   {PpT, "kW"}, {PpT / PaT, ""}, {SXR, ""}, {SIR, ""}, {SNR, ""}, {SDR, ""},
   {NS, ""}, {AeS NS, "km^2"}, {PaT AeT AeS NS, "meter^4*Watts"}};
prVals[solutions, ofInterest, {OmegaA → 0.01}, paramTot];
```

SBRmax = 5728.03

m = 12

TF = 0.4096 ms

AeS = 42 545.2 cm^2

TI = 2.17523 s

δ = 0.000188302

Ractual = 0.432 per second

PaT = 482.377 mW

AeT PaT = 0.00482377 m^2W

PpT = 10 492.8 kW

$\frac{\text{PpT}}{\text{PaT}} = 2.17523 \times 10^7$

SXR = 2441.41

SIR = 4.00723

SNR = 25 831.6

SDR = 336 033.

NS = 1432.01

AeS NS = 0.0060925 km^2

AeS AeT NS PaT = 29.3888 m^4W